Let us first define the 8 matrices as theu appear in the appendix.

```
In[1]:= L1 = {{0, 1, 0}, {1, 0, 0}, {0, 0, 0}};
        L2 = {{0, -I, 0}, {I, 0, 0}, {0, 0, 0}};
        L3 = {{1, 0, 0}, {0, -1, 0}, {0, 0, 0}};
        L4 = {{0, 0, 1}, {0, 0, 0}, {1, 0, 0}};
        L5 = {{0, 0, -I}, {0, 0, 0}, {I, 0, 0}};
        L6 = {{0, 0, 0}, {0, 0, 1}, {0, 1, 0}};
        L7 = {{0, 0, 0}, {0, 0, -I}, {0, I, 0}};
        L8 = {{1, 0, 0}, {0, 1, 0}, {0, 0, -2}} / Sqrt[3];
        L = {L1, L2, L3, L4, L5, L6, L7, L8};
```

```
In[10]:= Map[Print["L", #, " = ", MatrixForm[L[[#]]]] &, Range[8]];
```

$$L1 = \begin{pmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

$$L2 = \begin{pmatrix} 0 & -i & 0 \\ i & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

$$L3 = \begin{pmatrix} 1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

$$L4 = \begin{pmatrix} 0 & 0 & 1 \\ 0 & 0 & 0 \\ 1 & 0 & 0 \end{pmatrix}$$

$$L5 = \begin{pmatrix} 0 & 0 & -i \\ 0 & 0 & 0 \\ i & 0 & 0 \end{pmatrix}$$

$$L6 = \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & 1 & 0 \end{pmatrix}$$

$$L7 = \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & -i \\ 0 & i & 0 \end{pmatrix}$$

$$L8 = \begin{pmatrix} \frac{1}{\sqrt{3}} & 0 & 0 \\ 0 & \frac{1}{\sqrt{3}} & 0 \\ 0 & 0 & -\frac{2}{\sqrt{3}} \end{pmatrix}$$

Now let us compute the sum $\sum_{i=1}^{8} \mathrm{Li}^{\dagger}\, \mathrm{Li}$ to check that the mapping, with the normalization of 3/16, is indeed a trace preserving mapping. Since the Li's are Hermitian we can drop the dagger.

In[11]:=
```
Sum[(3 / 16) * L[[i]].L[[i]], {i, 8}] // MatrixForm
```

Out[11]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

So let us define the transformation as it appears in the text (with the corrected normalization factor).

In[12]:=
```
T[A_] := (3 / 16) * Sum[L[[i]].A.L[[i]], {i, 8}]
```

Let Id be the 3x3 identity matrix

In[13]:=
```
Id = IdentityMatrix[3]
```

Out[13]= {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}

Just checking, and indeed the transformation is trace preserving and unital.

In[14]:=
```
T[Id]
```

Out[14]= {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}

Let A be a generic 3x3 matrix

In[15]:=
```
A = Table[a[i, j], {i, 3}, {j, 3}];
```

Applying the channel to the matrix A we get

In[16]:=
```
Expand[T[A]] // MatrixForm
```

Out[16]//MatrixForm=

$$\begin{pmatrix} \frac{1}{4} a[1,1] + \frac{3}{8} a[2,2] + \frac{3}{8} a[3,3] & -\frac{1}{8} a[1,2] & -\frac{1}{8} a[1,3] \\ -\frac{1}{8} a[2,1] & \frac{3}{8} a[1,1] + \frac{1}{4} a[2,2] + \frac{3}{8} a[3,3] & -\frac{1}{8} a[2,3] \\ -\frac{1}{8} a[3,1] & -\frac{1}{8} a[3,2] & \frac{3}{8} a[1,1] + \frac{3}{8} a[2,2] + \frac{1}{4} a[3,3] \end{pmatrix}$$

Since for a quantum state A the trace is always 1, Tr[A]=1, we get that this is exactly $\frac{3}{8}$I- $\frac{1}{8}$A.
So just to verify we compute and indeed this is correct

In[17]:=
```
Expand[(3 / 8) * Tr[A] * IdentityMatrix[3] - A / 8] // MatrixForm
```

Out[17]//MatrixForm=

$$\begin{pmatrix} \frac{1}{4} a[1,1] + \frac{3}{8} a[2,2] + \frac{3}{8} a[3,3] & -\frac{1}{8} a[1,2] & -\frac{1}{8} a[1,3] \\ -\frac{1}{8} a[2,1] & \frac{3}{8} a[1,1] + \frac{1}{4} a[2,2] + \frac{3}{8} a[3,3] & -\frac{1}{8} a[2,3] \\ -\frac{1}{8} a[3,1] & -\frac{1}{8} a[3,2] & \frac{3}{8} a[1,1] + \frac{3}{8} a[2,2] + \frac{1}{4} a[3,3] \end{pmatrix}$$

In[18]:=
```
Tr[%]
```

Out[18]=
```
a[1, 1] + a[2, 2] + a[3, 3]
```

Now let us define three unital quantum channels S1, S2 and S3. The Kraus operators for S1 will be L1, L4 and L6; for S2 we will pick L2, L5 and L7; and for S3 we take L3 and L8.

In[19]:=
```
R1 = L[[{1, 4, 6}]];
R2 = L[[{2, 5, 7}]];
R3 = L[[{3, 8}]];
```

These are the Kraus operators for S1,

In[22]:=
```
Map[MatrixForm, R1]
```

Out[22]=

$$\left\{\begin{pmatrix}0&1&0\\1&0&0\\0&0&0\end{pmatrix},\begin{pmatrix}0&0&1\\0&0&0\\1&0&0\end{pmatrix},\begin{pmatrix}0&0&0\\0&0&1\\0&1&0\end{pmatrix}\right\}$$

these are the Kraus operators for S2,

In[23]:=
```
Map[MatrixForm, R2]
```

Out[23]=

$$\left\{\begin{pmatrix}0&-i&0\\i&0&0\\0&0&0\end{pmatrix},\begin{pmatrix}0&0&-i\\0&0&0\\i&0&0\end{pmatrix},\begin{pmatrix}0&0&0\\0&0&-i\\0&i&0\end{pmatrix}\right\}$$

and these are for S3.

In[24]:=
```
Map[MatrixForm, R3]
```

Out[24]=

$$\left\{\begin{pmatrix}1&0&0\\0&-1&0\\0&0&0\end{pmatrix},\begin{pmatrix}\frac{1}{\sqrt{3}}&0&0\\0&\frac{1}{\sqrt{3}}&0\\0&0&-\frac{2}{\sqrt{3}}\end{pmatrix}\right\}$$

Let us define the appropriate channels, each with the appropriate normalizing factor to make it trace preserving and unital.

In[25]:=
```
S1[A_] := (1 / 2) * Sum[R1[[i]].A.R1[[i]], {i, 3}]
S2[A_] := (1 / 2) * Sum[R2[[i]].A.R2[[i]], {i, 3}]
S3[A_] := (3 / 4) * Sum[R3[[i]].A.R3[[i]], {i, 2}]
```

Just checking the condition by applying the channels on the identity matrix

```
S1[Id]
```

```
{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}
```

In[28]:=
```
S2[Id]
```

Out[28]=
```
{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}
```

In[29]:=
```
S3[Id]
```

Out[29]=
```
{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}
```

Now we claim that the channel T is the convex combination of S1, S2 and S3, and that $T = \frac{3}{8}\text{S1} + \frac{3}{8}\text{S2} + \frac{2}{8}\text{S3}$. To check this we compute:

In[30]:=
```
MatrixForm[B = Expand[(3 / 8) * S1[A] + (3 / 8) * S2[A] + (2 / 8) * S3[A]]]
```

Out[30]//MatrixForm=

$$\begin{pmatrix} \frac{1}{4} a[1,1] + \frac{3}{8} a[2,2] + \frac{3}{8} a[3,3] & -\frac{1}{8} a[1,2] & -\frac{1}{8} a[1,3] \\ -\frac{1}{8} a[2,1] & \frac{3}{8} a[1,1] + \frac{1}{4} a[2,2] + \frac{3}{8} a[3,3] & -\frac{1}{8} a[2,3] \\ -\frac{1}{8} a[3,1] & -\frac{1}{8} a[3,2] & \frac{3}{8} a[1,1] + \frac{3}{8} a[2,2] + \frac{1}{4} a[3,3] \end{pmatrix}$$

and just to see that everything is OK lets compare this to the original channel T

In[31]:=
```
Expand[B - T[A]] // MatrixForm
```

Out[31]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

Thus we have shown that the channel is a convex combination of three other channels and as such cannot be an extremal channel.